# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python2-wtforms
_pkgname=WTForms
pkgver=2.2.1
pkgrel=5
pkgdesc='A flexible forms validation and rendering library for Python web development.'
url='https://pypi.org/project/WTForms/#files'
arch=('any')
license=('BSD')
depends=('python2')
makedepends=('python2-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('06478790a8a4bf728e5436edaa63c7c0e596f71d63b273e1f296f9255d9fa3e31a7bfa95971c214a8cea684034ac3c77b05371f4b4e14da9d9ef2c5afadabde0')

build() {
  cd "$_pkgname-$pkgver"

  python2 setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python2 setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build
}

